METODE NUMERICE DE REZOLVARE A ECUAȚIILOR ALGEBRICE ȘI TRANSCENDENTE

1. SEPARAREA SOLUTIILOR ECUATIILOR ALGEBRICE ŞI TRANSCENDENTE

Ne amintim!

Dacă funcția $f(x)$ are forma unui polinom sau poate fi adusă la această formă, ecuația $f(x) = 0$ se numește ***algebrică***.

Exemplu: $3x - 7 = 0$.

În caz contrar – cînd $f(x)$ nu este una polinomială –, ecuația se numește ***transcendentă***.

Exemplu: $\sin(2x) + \sqrt{\cos^2 x + e^x} = 0$.

Orice valoare ξ, pentru care expresia f(ξ) = 0 este adevărată, se numeşte **zerou** al functiei f(x) sau **solutie** a ecuatiei f(x) = 0.

Rezolvarea unei ecuatii algebrice se divide în două etape:

- 1. Separarea intervalelor pe care ecuatia are o singură solutie
- 2. Micşorarea pe cît mai mult posibil a fiecărui din aceste intervale (dacă se pune problema determinării tuturor solutiilor) sau a unuia din ele (dacă trebuie de determinat doar una din solutii).

Pentru separarea solutiilor se va folosi următoarea teoremă din analiza matematică:

Teoremă.
Dacă funcţia f(x), continuă pe segmentul [a, b], primeşte la extremităţile lui valori de semn diferit ($f(a) \times f(b) < 0$), atunci pe acest segment există cel puţin un punct ξ, astfel încît $f(\xi) = 0$.
Dacă pe [a, b] există derivata $f'(x)$, continuă, care are un semn constant, atunci ξ este soluţie unică a ecuaţiei $f(x) = 0$ pe acest segment.

Metoda analitică.

Exemplul 1: Să se determine numărul de soluții ale ecuației $e^x + x = 0$

$$f'(x) = e^x + 1; \quad f'(x) > 0 \ \forall \ x \in R.$$

Întrucît $\lim\limits_{x \to -\infty} f(x) = -\infty$, $\lim\limits_{x \to \infty} f(x) = \infty$, ecuația inițială are o singură soluție.

Exemplul 2: Să se separe rădăcinile ecuației $x^3 - 9x^2 + 24x - 19 = 0$ pe segmentul $[0, 8]$.

Rezolvare:

$$f(x) = x^3 - 9x^2 + 24x - 19;$$
$$f'(x) = 3x^2 - 18x + 24.$$

Rezolvînd ecuația $f'(x) = 0$, se obțin soluțiile $x_1 = 2$, $x_2 = 4$.

x	$f(x)$	Semn $f(x)$
0	−19	−
2	1	+
4	−3	−
8	109	+

Deci ecuația va avea trei soluții, cîte una pe fiecare din segmentele $[0, 2]$, $[2, 4]$, $[4, 8]$.

Metoda grafică.

Separarea grafică a soluțiilor unei ecuații pe un segment dat poate fi realizată și local, cu ajutorul unei aplicații de calcul tabelar. Este suficient să se construiască un tabel cu două coloane. Prima coloană va reprezenta o divizare a segmentului în segmente elementare de lungimi egale. Cea de-a doua coloană va conține o formulă care calculează valoarea funcției f(x) pentru valorile respective din prima coloană. În baza datelor din coloana cu valorile f(x) se construiește o diagramă liniară, care reprezintă graficul funcției analizate.

Exemplu: $f(x) = x^{\cos(2x)} + 3\sin(x)$. Soluțiile se caută pe segmentul [0,2, 10] (*fig. 3.1 a, 3.1 b*).

B2 =EXP(COS(2*A2)*LN(A2))+3*SIN(A2)

	A	B	C	D	E	F
1	x	y				
2	0,2	0,823102167				
3	0,4	1,696399241				
4	0,6	2,524947247				
46	9	5,503155321				
47	9,2	8,048154414				
48	9,4	9,448504359				
49	9,6	7,844007308				
50	9,8	4,209027748				
51	10	0,927006056				

Fig. 3.1a. Funcția *f*(*x*) reprezentată tabelar în foaia de calcul. Colana A conține valorile *x* de la 0,2 la 10 cu *pasul 0,2*. Coloana B conține valorile *f*(*x*)



Fig. 3.1b. Funcția *f*(*x*) reprezentată grafic în baza datelor din tabel. Pot fi ușor identificate două segmente oarecare, ce conțin exact cîte o soluție a ecuației *f*(*x*)=0, de exemplu: [3, 5] și [5, 6]

Metoda bisecţiei

Una dintre cele mai simple metode de determinare a unei soluţii a ecuaţiei f(x) = 0este metoda bisecţiei.

Metoda presupune determinarea punctului de mijloc c al segmentului [a, b], apoi calculul valorii f(c).
Dacă f(c) = 0, atunci c este soluţia exactă a ecuaţiei.
În caz contrar, soluţia este căutată pe unul dintre segmentele [a, c] şi [c, b].
Ea va aparţine segmentului pentru care semnul funcţiei în extremităţi este diferit.



Fig. 3.2. Calculul consecutiv al segmentelor, care conţin soluţia ecuaţiei *f*(*x*)=0

ALGORITMIZAREA METODEI

Pornind de la descrierea matematică a metodei, putem separa două cazuri distincte de oprire a procesului de calcul al soluției ecuației $f(x) = 0$ pentru metoda bisecției:

A1. *Algoritmul de calcul pentru un număr prestabilit n de divizări consecutive:*

Pasul 0. *Inițializare:* $i \Leftarrow 0$.

Pasul 1. Determinarea mijlocului segmentului $c \Leftarrow \frac{a+b}{2}$.

Pasul 2. Reducerea segmentului ce conține soluția: dacă $f(c) = 0$, atunci soluția calculată este $x = c$. SFÎRȘIT.
În caz contrar, dacă $f(a) \times f(c) > 0$, atunci $a \Leftarrow c$; $b \Leftarrow b$, altfel $a \Leftarrow a$; $b \Leftarrow c$.

Pasul 3. $i \Leftarrow i + 1$. Dacă $i = n$, atunci soluția calculată este $x = \frac{a+b}{2}$. SFÎRȘIT.
În caz contrar, se revine la *pasul 1*.

A2. *Algoritmul de calcul pentru o precizie[2] ε dată:*

Pasul 1. Determinarea mijlocului segmentului $c \Leftarrow \frac{a+b}{2}$.

Pasul 2. Dacă $f(c) = 0$, atunci soluția calculată este $x = c$. SFÎRȘIT.
În caz contrar, dacă $f(a) \times f(c) > 0$, atunci $a \Leftarrow c$; $b \Leftarrow b$, altfel $a \Leftarrow a$; $b \Leftarrow c$.

Pasul 3. Dacă $|b - a| < \varepsilon$, atunci soluția calculată este $x = \frac{a+b}{2}$. SFÎRȘIT.
În caz contrar, se revine la *pasul 1*.

În caz contrar, se revine la pasul 1.

Exemplul 1: Să se determine o rădăcină a ecuației $x^4 + 2x^3 - x - 1 = 0$ pe segmentul $[0, 1]$ pentru 16 divizări consecutive.

Deoarece numărul de aproximări succesive este fixat, iar extremitățile segmentului cunoscute, atribuirile se realizează nemijlocit în program.

```
program cn05;
var   a,b,c: real;
      i,n:integer;

function f(x:real):real;
begin f:=sqr(sqr(x))+2*x*sqr(x)-x-1;end;
begin a:=0; b:=1; n:=16;
      for i:=1 to n do
        begin c:=(b+a)/2;
          writeln('i=',i:3,' x=',c:10:8,' f(x)=',f(c):12:8);
          if f(c)=0 then break[3]
          else if f(c)*f(a)>0 then a:=c else b:=c;
        end;
end.
```

Rezultate:

```
i= 1  x=0.50000000 f(x)= -1.18750000
i= 2  x=0.75000000 f(x)= -0.58984375
...
i= 15 x=0.86679077 f(x)= 0.00018565
i= 16 x=0.86677551 f(x)= 0.00009238
```

Metoda coardelor

Metoda bisecţiei, cu toată simplitatea ei, nu este eficientă în cazurile cînd rezultatul trebuie obţinut printr-un număr redus de iteraţii, cu o exactitate înaltă. Astfel stînd lucrurile, este mai potrivită ***metoda coardelor***, care constă în divizarea segmentului în părţi proporţionale, proporţia fiind dată de punctul de intersecţie al coardei care uneşte extremităţile segmentului cu axa ox.

Metoda coardelor presupune alegerea în calitate de aproximare a soluției punctul determinat de intersecția dreptei ce trece prin punctele (a, f(a)) și (b, f(b)) cu axa Ox.





Fig. 3.3. Apropierea succesivă de soluția ecuației prin metoda coardelor

ALGORITMIZAREA METODEI

A1. ***Algoritmul de calcul pentru un număr prestabilit n de aproximări succesive:***

Pasul 1. Determinarea extremității fixe e și a aproximării x_0:

$$c \Leftarrow a - \frac{f(a)}{f(b) - f(a)}(b - a);$$

dacă $f(c) \times f(a) < 0$, atunci $e \Leftarrow a$, $x_0 \Leftarrow b$, altfel $e \Leftarrow b$, $x_0 \Leftarrow a$; $i \Leftarrow 0$.

Pasul 2. Calculul x_{i+1} conform formulei $x_{i+1} = x_i - \dfrac{f(x_i)}{f(e) - f(x_i)}(e - x_i)$.

Pasul 3. Dacă $i + 1 = n$, atunci soluția calculată $x \Leftarrow x_i$. SFÎRȘIT.
În caz contrar, $i \Leftarrow i+1$ și se revine la *pasul 2*.

ALGORITMIZAREA METODEI

A2. ***Algoritmul de calcul pentru o exactitate ε dată:***

Deoarece în formula de estimare a erorii figurează mărimile M_1 și m_1, atunci cînd valorile lor nu sînt indicate în enunțul problemei, este necesară descrierea analitică a $f'(x)$ și calcularea M_1 și m_1.

Pasul 1. Determinarea extremității fixe e și a aproximării x_0:

$$c \Leftarrow a - \frac{f(a)}{f(b) - f(a)}(b - a);$$

dacă $f(c) \times f(a) < 0$, atunci $e \Leftarrow a$, $x_0 \Leftarrow b$, altfel $e \Leftarrow b$, $x_0 \Leftarrow a$; $i \Leftarrow 0$.

Pasul 2. Calculul x_{i+1} conform formulei $x_{i+1} = x_i - \dfrac{f(x_i)}{f(e) - f(x_i)}(e - x_i)$.

Pasul 3. Dacă $\left|\dfrac{M_1 - m_1}{m_1}\right| \times \left|x_{i+1} - x_i\right| \le \varepsilon$, atunci soluția calculată $x \Leftarrow x_i$. SFÎRȘIT.

În caz contrar, $i \Leftarrow i+1$ și se revine la *pasul 2*.

Exemplul 2: Fie dată funcția $f(x) = x^4 - 3x^2 + 7{,}5x - 1$. Să se calculeze soluția aproximativă a ecuației $f(x) = 0$ pe segmentul $[-0{,}5; 0{,}5]$ cu exactitatea $\varepsilon = 0{,}0001$, utilizînd metoda coardelor. Pentru funcția dată pe $[-0{,}5; 0{,}5]$ M_1 și m_1 sînt, respectiv, egale cu 10 și 5.

Pentru simplitate, atribuirile necesare vor fi realizate direct în corpul programului.

```
program cn08;
var
  Msup,minf,a,b,e,x,xnou,xvechi,eps: real;
function f(x:real):real;
begin
  f:=sqr(sqr(x))-3*sqr(x)+7.5*x-1;
end;
begin
  a:=-0.5; b:=0.5; eps:=0.0001;
  Msup:=10; minf:=5;
  {determinarea extremitatii fixe si a aproximarii initiale}
  x:=a-(f(a))/(f(b)-f(a))*(b-a);
  if f(x)*f(a)>0 then begin e:=b; xnou:=a; end
  else begin e:=a; xnou:=b; end;
  {calculul iterativ al solutiei}
  repeat
  xvechi:=xnou;
  xnou:= xvechi-(f(xvechi))/(f(e)-f(xvechi))*(e-xvechi);
  writeln(' x=',xnou:10:8,' f(x)=',f(xnou):12:8);
  until abs((Msup-minf)/minf*(xnou-xvechi))<eps;
end.
```

Metoda Newton

La fel ca în paragrafele precedente, scopul este de a rezolva ecuatia f(x) = 0 pentru x ∈ [a, b].

Se va încerca rezolvarea problemei prin trasarea consecutivă a unor tangente la graficul funcției, prima dintre ele fiind construită prin extremitatea $E_0 (x_0, y_0)$ a segmentului $[a, b]$, extremitate pentru care se respectă condiția: $f(x_0) \times f''(x_0) > 0$.

Fie că tangenta cu numărul i intersectează axa $0x$ în punctul x_i (*fig. 3.4*). Următoarea tangentă $(i+1)$ va fi trasată prin punctul E_{i+1} cu coordonatele $(x_i, f(x_i))$ și va intersecta axa absciselor în punctul x_{i+1}. Șirul de valori $x_0, x_1, x_2, ..., x_i, x_{i+1}, ...$ va converge către soluția ecuației $f(x) = 0$. Această metodă de calcul al soluției ecuației $f(x)=0$ este numită *metoda tangentelor* sau *Newton*, după numele matematicianului care a introdus-o.



Fig. 3.4. Convergența şirului de valori $x_0, x_1, x_2, ..., x_i, x_{i+1}, ...$ către soluția exactă ξ

ALGORITMIZAREA METODEI

Numărul de aproximări succesive în procesul de calcul poate fi stabilit apriori sau determinat de o condiție. Mai întîi se stabilește extremitatea segmentului care va servi drept aproximare inițială. Calculul aproximării următoare se realizează în ambele cazuri conform formulei (6). Condiția de oprire a calculelor va fi în primul caz generarea aproximării cu indicele cerut; în cel de al doilea – îndeplinirea condiției (7).

A1. ***Algoritmul de calcul pentru un număr dat de aproximări succesive:***

Pentru a realiza acest algoritm, este suficient să fie cunoscute descrierile analitice pentru $f(x)$ și $f'(x)$. Dacă descrierea $f'(x)$ nu este indicată în enunț, urmează să fie calculată. Aproximarea inițială se deduce utilizînd procedeul similar determinării extremității fixe pentru metoda coardelor.

Pasul 1. Determinarea aproximării inițiale x_0: $c \Leftarrow a - \dfrac{f(a)}{f(b) - f(a)}(b - a)$;

dacă $f(c) \times f(a) < 0$, atunci $x_0 \Leftarrow a$, altfel $x_0 \Leftarrow b$; $i \Leftarrow 0$.

Pasul 2. Se calculează x_{i+1} conform formulei $x_{i+1} = x_i - \dfrac{f(x_i)}{f'(x_i)}$.

Pasul 3. Dacă $i+1 = n$, atunci soluția calculată $x \Leftarrow x_{i+1}$. SFÎRȘIT.

În caz contrar, $i \Leftarrow i+1$, apoi se revine la *pasul 2*.

ALGORITMIZAREA METODEI

A2. *Algoritmul de calcul pentru o exactitate ε dată:*

În formula de estimare a erorii figurează mărimile M_2 și m_1. Atunci cînd valorile lor nu sînt indicate în enunțul problemei, este necesară o preprocesare matematică pentru stabilirea M_2 și m_1. Suplimentar sînt necesare descrierile analitice pentru $f(x)$ și $f'(x)$.

Pasul 1. Determinarea aproximării inițiale x_0: $c \Leftarrow a - \dfrac{f(a)}{f(b)-f(a)}(b-a)$;

dacă $f(c) \times f(a) < 0$, atunci $x_0 \Leftarrow a$, altfel $x_0 \Leftarrow b$; $i \Leftarrow 0$.

Pasul 2. Se calculează x_{i+1} conform formulei $x_{i+1} = x_i - \dfrac{f(x_i)}{f'(x_i)}$.

Pasul 3. Dacă $\dfrac{M_2}{2m_1}(x_{i+1} - x_i)^2 \le \varepsilon$, atunci soluția calculată $x \Leftarrow x_{i+1}$. SFÎRȘIT.

În caz contrar, $i \Leftarrow i+1$ și se revine la *pasul 2*.

Exemplul 1[3]**:** Fie dată funcția $f(x) = x^3 - 2x^2 + x - 3$. Să se scrie un program care va calcula soluția ecuației $f(x) = 0$ pe segmentul [2; 15] pentru 10 aproximări succesive, utilizînd metoda Newton.

Preprocesarea matematică: Se determină $f'(x)$:

$$f(x) = x^3 - 2x^2 + x - 3; \qquad f'(x) = 3x^2 - 4x + 1.$$

Deoarece numărul de aproximări succesive este fixat, iar extremitățile segmentului cunoscute, atribuirile necesare se vor realiza direct în corpul programului.

```
program cn09;
var a, b, x, c : real;
   i, n: integer;
function f(z:real):real;
     begin f:=z*z*z-2*z*z+z-3; end;
function fd1(z:real):real;
     begin fd1:=3*z*z-4*z+1; end;
begin   a:=2.1; b:=15; n:=10; i:=0;
        c:=a-(f(a))/(f(b)-f(a))*(b-a);
        if f(c)*f(a)<0 then x:=a else x:=b;
        while i<n do
        begin   i:=i+1;
                x:=x-f(x)/fd1(x);
        writeln('i=',i:2,' x=',x:15:12, ' f=',f(x):15:12);
        end;
end.
```

Rezultate:

```
i= 1 x= 10.23214285700  f=869.11072454000
i= 2 x= 7.06207637180   f=256.52261987000
...
i= 9 x= 2.17455942470   f= 0.00000009329
i=10 x= 2.17455941030   f= 0.00000000001
```

Proiect realizat de:
Sandu Evelina cl. 12 “B”
IPLT “Spiru Haret”
Prof. Maria Gutu
Noiembrie 2018

Biografie:

- Informatică: Manual pentru clasa a 12-a/Anatol Gremalschi, Sergiu Corlat, Andrei Braicov; Min. Educaţiei al Rep. Moldova. – Ch.: Î.E.P. Ştiinţa, 2015 (Tipografia „BALACRON” SRL) – 144 p. (http://www.ctice.md/ctice2013/?page_id=1690)
- http://www.math.md/stireal/informatica/candidat/calcul_numeric_3.pdf